Using LQR to stabilize an Inverted pendulum

Nasser M. Abbasi
April 12, 2012



Introduction

This is an analysis of the dynamics of inverted bob pendulum on a moving cart. The equations of motion for the cart and the pendulum bob mass are derived using Lagrangian formulation, then state space model is derived and then LQR is used to find the state gain vector to bring the pendulum to the upright position from an initial position. The analysis uses *Mathematica*. Viscous friction is assumed to be present. This is friction between the cart itself and the rail that the cart moves on.

Derivation of the equations of motion

Let ν be the viscous friction coefficient. The potential energy of the system is PE $= m\,g\,L\sin(\theta)$ and the kinetic energy is KE $= \frac{1}{2} m \left((x' - L\,\theta' \sin\theta)^2 + (L\,\theta' \cos\theta)^2 \right) + \frac{1}{2} M\,(x')^2$, hence the Lagrangian is now found and equations of motion is found for the bob mass and for the cart mass as follows

```
Clear[m, M, len, θ, x, t, ν, L]
ke = (1 / 2) m ((x'[t] - len * θ'[t] Sin[θ[t]]) ^2 + (len * θ'[t] Cos[θ[t]]) ^2) +
   1 / 2 M * x'[t] ^2;
pe = m * g * len * Sin[θ[t]];
```

The Lagrangian

```
L = ke - pe
```

$$-g\,\text{len}\,m\,\text{Sin}[\theta[t]]+\frac{1}{2}M\,x'[t]^2+\frac{1}{2}m\left(\text{len}^2\,\text{Cos}[\theta[t]]^2\,\theta'[t]^2+(x'[t]-\text{len}\,\text{Sin}[\theta[t]]\,\theta'[t])^2\right)$$

find equation of motion for the bob

```
eq1 = D[D[L, θ'[t]], t] - D[L, θ[t]] == 0;
eq1 = Simplify[eq1]
```

$$\text{len}\,m\,(g\,\text{Cos}[\theta[t]]-\text{Sin}[\theta[t]]\,x''[t]+\text{len}\,\theta''[t])==0$$

find equation of motion for the cart.

```
eq2 = D[D[L, x'[t]], t] - D[L, x[t]] == f[t] - ν x'[t];
eq2 = Simplify[eq2]
```

$$\nu\,x'[t]+(m+M)\,x''[t]==f[t]+\text{len}\,m\left(\text{Cos}[\theta[t]]\,\theta'[t]^2+\text{Sin}[\theta[t]]\,\theta''[t]\right)$$

Find the state space representation using 90 degree and x=0 as the equilibrium point

```
sys = StateSpaceModel[{eq1, eq2},
  {{x[t], 0}, {θ[t], Pi / 2}, {x'[t], 0}, {θ'[t], 0}}, {{f[t], 0}}, {x[t], θ[t]}, t]
```

$$\left(\begin{array}{cccc|c}
0 & 0 & 1 & 0 & 0\\
0 & 0 & 0 & 1 & 0\\
0 & \frac{g\,m}{M} & -\frac{\nu}{M} & 0 & \frac{1}{M}\\
0 & \frac{g\,(m+M)}{\text{len}\,M} & -\frac{\nu}{\text{len}\,M} & 0 & \frac{1}{\text{len}\,M}\\
\hline
1 & 0 & 0 & 0 & 0\\
0 & 1 & 0 & 0 & 0
\end{array}\right)^{\mathcal{S}}$$

Hence, the A matrix is

```
sys[[1, 1]] // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & \frac{g\,m}{M} & -\frac{\nu}{M} & 0 \\ 0 & \frac{g\,(m+M)}{\text{len}\,M} & -\frac{\nu}{\text{len}\,M} & 0 \end{pmatrix}$$

and the B matrix is

```
sys[[1, 2]] // MatrixForm
```

$$\begin{pmatrix} 0 \\ 0 \\ \frac{1}{M} \\ \frac{1}{\text{len}\,M} \end{pmatrix}$$

And the C matrix is

```
sys[[1, 3]] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{pmatrix}$$

obtain optimal state feedback gain. First make up a weight matrix Q

```
values = {m → 1, M → 10, len → 5, g → 9.8, ν → 2};
(q = DiagonalMatrix[{100, 1, 1, 1}]) // MatrixForm
```

$$\begin{pmatrix} 100 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Now use LQR to find state feedback gain vector

```
gain = LQRegulatorGains[sys /. values, {q, {{1}}}]
```

```
{{-10., 514.055, -30.8807, 360.115}}
```

Plot the response of the system using this state feedback gain from an initial position. First generate the closed loop system

```
closedLoopSys = SystemsModelStateFeedbackConnect[sys, gain]
```

$$
\left(\begin{array}{cccc|c}
0. & 0. & 1. & 0. & 0. \\
0. & 0. & 0. & 1. & 0. \\
0. + \frac{10.}{M} & 0. - \frac{514.055}{M} + \frac{g\,m}{M} & 0. + \frac{30.8807}{M} - \frac{\nu}{M} & 0. - \frac{360.115}{M} & \frac{1}{M} \\
0. + \frac{10.}{\text{len}\,M} & 0. - \frac{514.055}{\text{len}\,M} + \frac{g\,(m+M)}{\text{len}\,M} & 0. + \frac{30.8807}{\text{len}\,M} - \frac{\nu}{\text{len}\,M} & 0. - \frac{360.115}{\text{len}\,M} & \frac{1}{\text{len}\,M} \\
\hline
1. & 0. & 0. & 0. & 0. \\
0. & 1. & 0. & 0. & 0.
\end{array}\right)^{\mathcal{S}}
$$

```
initialAngle = 75 Degree;
states =
 Chop[StateResponse[{closedLoopSys /. values, {2, initialAngle, 0, 0}}, 0, t]]
```

$$
\begin{aligned}
\{ & e^{-2.15712\,t} \left(-56.1216\, e^{0.698918\,t} \,\text{Cos}[0.0247103\,t] + 58.1216\, e^{1.4582\,t}\, \text{Cos}[0.658042\,t] + \right. \\
& \quad \left. 147.687\, e^{0.698918\,t}\, \text{Sin}[0.0247103\,t] - 68.1774\, e^{1.4582\,t}\, \text{Sin}[0.658042\,t]\right), \\
& e^{-2.15712\,t} \left(4.88032\, e^{0.698918\,t}\, \text{Cos}[0.0247103\,t] - 3.57133\, e^{1.4582\,t}\, \text{Cos}[0.658042\,t] + \right. \\
& \quad \left. 371.85\, e^{0.698918\,t}\, \text{Sin}[0.0247103\,t] - 6.94194\, e^{1.4582\,t}\, \text{Sin}[0.658042\,t]\right), \\
& e^{-2.15712\,t} \left(85.4858\, e^{0.698918\,t}\, \text{Cos}[0.0247103\,t] - 85.4858\, e^{1.4582\,t}\, \text{Cos}[0.658042\,t] - \right. \\
& \quad \left. 213.97\, e^{0.698918\,t}\, \text{Sin}[0.0247103\,t] + 9.40396\, e^{1.4582\,t}\, \text{Sin}[0.658042\,t]\right), \\
& e^{-2.15712\,t} \left(2.07203\, e^{0.698918\,t}\, \text{Cos}[0.0247103\,t] - 2.07203\, e^{1.4582\,t}\, \text{Cos}[0.658042\,t] - \right. \\
& \quad \left. 542.352\, e^{0.698918\,t}\, \text{Sin}[0.0247103\,t] + 7.20193\, e^{1.4582\,t}\, \text{Sin}[0.658042\,t]\right) \}
\end{aligned}
$$

Plot the x(t) and θ(t) responses to see if they do come to the equilibrium position

```
maxt = 10;
Row[{Plot[states[[1]], {t, 0, maxt}, PlotLabel → "x(t) vs t",
   ImageSize → 300, PlotRange → All], Plot[states[[2]], {t, 0, maxt},
   PlotLabel → "θ(t) vs t", ImageSize → 300, PlotRange → All]}]
```



We see from above that both the x position of the cart and the upright pendulum position have been brought back to the equilibrium position